

# **A WAXICON Design Code for Arbitrary Input-Output Intensity Distributions**

Eugene A. Aronson, Louis A. Romero, Alexander H. Treadway

Prepared by  
Sandia National Laboratories  
Albuquerque, New Mexico 87185 and Livermore, California 94550  
for the United States Department of Energy  
under Contract DE-AC04-76DP00789



SAND86-0770  
Unlimited Release  
Printed xxxx 1986

A WAXICON DESIGN CODE FOR ARBITRARY  
INPUT-OUTPUT INTENSITY DISTRIBUTIONS

Eugene A. Aronson,  
Louis A. Romero,  
Alexander H. Treadway  
Applied Mathematics Division 2646  
Sandia Laboratories, Albuquerque, NM 87185

ABSTRACT

A method and computer code are described for the design of a WAXICON of arbitrary input and output intensity distributions. The source may emanate from a point, which may be at infinity.

## CONTENTS

	<u>Page</u>
Introduction	1
Geometry and Differential Equations	2
Numerical Considerations	4
The Code WXIO	5
PROGRAM WXIO Listing	7

## Introduction

A WAXICON is a device for redirecting and reshaping light intensity with mirrors rather than lenses. In this report the discussion is restricted to the axially symmetric WAXICON whose input-output rays are  $180^\circ$  apart. Refer to Figure 1. We use a cylindrical coordinate system with  $r$  as the radial dimension and  $z$  as the axial. The WAXICON mirrors are radially symmetric about the  $z$ -axis. A spherically symmetric cone of light emanates from the point  $(0, -L)$ . This point is allowed to be at  $(0, -\infty)$ . A typical ray reflects off the inner (input) mirror at the point  $(X, F)$  and then off the outer (output) mirror at  $(Y, G)$ . All output rays must be parallel to the  $z$ -axis and moving in the minus  $z$  direction. Also, the relative input-output phase front must be preserved.

The code WXIO designs the mirrors so that an arbitrary output light intensity distribution is achieved for a given input intensity distribution. The designer also specifies the overall dimensions: the radius of the inner mirror,  $A$ , the inner radius of the output mirror,  $B$ , the outer radius of the outer mirror,  $C$ , and the translational offset of the outer mirror relative to the inner mirror,  $D$ . The code is written in FORTRAN77 and is available on any computer system supporting this compiler.

## Geometry and Differential Equations

From Fermat's principle we can show that any reflecting optical system which focuses rays emerging from one point onto another point must preserve phase. Either, neither, or both of these points may be at infinity. For our purpose, all rays are of the same length, and thus preserve phase, if they emanate from a point, reflect off the inner mirror, then reflect off the outer mirror such that they are parallel (i. e., focus at infinity).

Again, refer to Figure 1. The center of the input mirror is at  $(0, 0)$ . The mirror radii are given as  $0 < A < B < C$ . The offset,  $D$ , may be of any sign. The light source is at  $(0, -L)$ , where  $L > 0$  may be at infinity. A ray from the source towards the inner mirror crosses the  $r$ -axis at the point  $(R, 0)$ , impinges on the inner mirror at  $(X, F)$  and on the outer mirror at  $(Y, G)$ . We use  $R$  as the independent variable in the  $r$  direction. The input cone of light is terminated at the rays which hit the inner mirror at  $r = A$ . These input rays cross the  $r$ -axis at  $r = T$ .

Since each ray must have the same length, to preserve phase, we get

$$\sqrt{X^2 + (F+L)^2} + \sqrt{(Y-X)^2 + (G-F)^2} + G = K + L, \quad (1)$$

where the constant  $K$  is

$$K = \sqrt{D^2 + C^2} + D. \quad (2)$$

An arbitrary segment of the ray going from  $(X, F)$  to  $(Y, G)$  has the equation

$$z = S(r - X) + F,$$




Figure 1.




Figure 2.

where S is its slope. Since this line segment intersects the point (Y,G),

$$G = S(Y - X) + F. \quad (3)$$

Substituting equation (3) in equation (1) gives the quantity H as

$$H = S + \sqrt{1 + S^2} = [K + L - F - \sqrt{X^2 + (F+L)^2}]/(Y - X), \quad (4)$$

and

$$S^2 = (H^2 - 1)/2H. \quad (5)$$

Refer to Figure 2. The angles a, f, and g are defined as

$$\tan a = X/(F + L) = R/L,$$

$$\tan f = dF/dX,$$

$$\tan g = -dG/dY.$$

We have

$$e + f + a + 90^\circ = 180^\circ \rightarrow e = 90^\circ - f - a,$$

$$2e + a + b + 90^\circ = 180^\circ \rightarrow b = 2f + (a - 90^\circ).$$

Now,

$$\begin{aligned} S = \tan b &= [\tan 2f + \tan(a-90^\circ)]/[1 - \tan 2f \tan(a-90^\circ)] \\ &= (\tan 2f \tan a - 1)/(\tan 2f + \tan a) \end{aligned} \quad (6)$$

And,

$$dF/dX = \tan\{\arctan[(S \tan a + 1)/(\tan a - S)]/2\}, \quad (7)$$

$$X = R(F/L + 1). \quad (8)$$

Therefore,

$$\frac{dF}{dR} = \frac{(dF/dX)(F/L + 1)}{1 - (dF/dX)R/L}. \quad (9)$$

Let the input intensity distribution (energy per unit area) be given by  $p(r) > 0$ . We specify  $p(r)$  in the disk at  $z = 0$  and  $0 \leq r \leq T$ . It is desired to find  $F(X)$  and  $G(Y)$  so that the output intensity distribution "shape" is the specified function  $q(r)$  for  $B \leq r \leq C$ . Since no energy is lost,

$$E \int_Y^C 2\pi r q(r) dr = \int_0^R 2\pi r p(r) dr, \quad (10)$$

where E is the scale factor given by

$$E = \int_0^T rp(r)dr / \int_B^C rq(r)dr . \quad (11)$$

Differentiating equation (10) with respect to R gives

$$dY/dR = -Rp(R)/[EYq(Y)]. \quad (12)$$

Equations (9) and (12) are a pair of simultaneous differential equations with  $0 \leq R \leq T$  as the independent variable, F and Y as dependent variables, and initial conditions  $F(0) = 0$  and  $Y(0) = C$ . The code WXIO numerically solves these equations.

### Numerical Considerations

Since we allow L to go to infinity there is a difficulty in evaluating H, equation (4). We confine the ratio A/L to be less than 1/10, not restrictive in a practical situation. Using the binomial expansion H is approximated by

$$H = [K - 2F - X \tan a \left( \frac{1}{2} - \frac{\tan^2 a}{8} + \frac{\tan^4 a}{16} - \frac{5\tan^6 a}{128} + \frac{7\tan^8 a}{256} \right)] / (Y - X).$$

Note that all expressions are now evaluated in terms of 1/L, which goes to zero as L goes to infinity.

In order to solve the differential equations it is necessary to know the value of T. However, T is unknown because the value of F at  $X = A$  is not known a priori, unless  $1/L = 0$ , in which case  $T = A$ . We find T by Newton's method. As a first guess we choose  $T_0 = 0.98A/(A/L + 1)$ . The 0.98 is a safety factor since we wish to approach the final T from below. Successive runs are made and the value of T is corrected by the iteration

$$T_{i+1} = T_i - (X - A)/(dX/dR), \quad (13)$$

where X and  $dX/dR$  are evaluated at  $R = T_i$ . The derivative  $dX/dR$  is

$$dX/dR = R(dF/dR)/L + F/L + 1.$$

An internal test is found by reference to Figure 2. For the exit ray to be parallel to the z-axis it is required that

$$2(90^\circ - g) + 90^\circ - b = 180^\circ \rightarrow g = 90^\circ - f - a/2.$$

Thus as a check we need

$$-\tan(dG/dY) = 90^\circ - \tan(dF/dX) - a/2.$$

We have

$$dG/dY = (dG/dR)/(dY/dR) = [(dS/dR)(Y-X) + S(dY/dR - dX/dR) + dF/dR]/(dY/dR),$$

$$dS/dR = (H^2 + 1)(dH/dR)/2H,$$

$$dH/dR = [H(dX/dR - dY/dR) - dF/dR - Q]/(Y - X),$$

where

$$Q = [X(dX/dR)/L + (F/L+1)(dF/dR)]/\sqrt{(X/L)^2 + (F/L+1)^2}.$$

The formulas above were used to check the results, which indicated that the code is correct.

### The Code WXIO

The user supplies values for A, B, C, D, 1/L, and NP, where NP > 4 is the number of points to be printed in the range  $0 \leq R \leq T$  ( $0 \leq X \leq A$ ). Any viable input or output intensity function can be used by modifying the FUNCTION subprograms PEV and QEV. At present the only output option included is UNIFORM. The current input options are:

1. UNIFORM      $p(r) = 1, \quad (0 \leq r \leq T),$
2. COSINE      $p(r) = \cos(\pi r/2T)$
3. GAUSS-1     $p(r) = \exp[-(r/T)^2/2]$
4. GAUSS-2     $p(r) = \exp[-(2r/T)^2/2]$
5. GAUSS-3     $p(r) = \exp[-(3r/T)^2/2].$

The code is written in FORTRAN77 and is available for any computer system with that compiler. The solution of the differential equations is performed by the SLATEC library routine DERKF, which is a Runge-Kutta type of routine. A short test run is listed and plotted below.

WXIO	A	B	C	D	1/L	NP
	0.3440	0.5850	1.7500	-0.2500	0.0900	11
INPUT:GAUSS-2		OUTPUT:UNIFORM				
	X	F	Y	G		
	0.00000	0.25000	1.75000	0.00000		
	0.03324	0.27861	1.73211	0.02078		
	0.06664	0.30731	1.67947	0.08172		
	0.10023	0.33674	1.59507	0.17725		
	0.13400	0.36757	1.48356	0.29805		
	0.16800	0.40065	1.35079	0.43265		
	0.20225	0.43705	1.20324	0.56922		
	0.23682	0.47834	1.04747	0.69726		
	0.27181	0.52692	0.88947	0.80870		
	0.30740	0.58714	0.73420	0.89831		
	0.34400	0.66902	0.58499	0.96327		




Figure 3. Plot of Sample Run.

C PROGRAM WXIO(TAPE6, INPUT, OUTPUT)  
 C \*\*\*\*\*  
 C \* E. A. ARONSON, L. A. ROMERO, A. H. TREADWAY \*  
 C \* DIVISION 2646 SANDIA NATIONAL LABS \*  
 C \* ALBUQUERQUE, NM 87185 \*  
 C \* MARCH 19, 1986 \*

C \*\*\*\*\*  
 C \* \* \* \* \*  
 C \* ISSUED BY SANDIA \*  
 C \* NATIONAL LABORATORIES, \*  
 C \* OPERATED FOR THE \*  
 C \* \* \* \* \* UNITED STATES \*  
 C \* DEPARTMENT OF \*  
 C \* ENERGY BY \*  
 C \* SANDIA \*  
 C \* CORPORATION \*

C \*\*\*\*\* ---NOTICE--- \*\*\*\*\*  
 C \*THIS REPORT WAS PREPARED AS AN ACCOUNT OF WORK SPONSORED\*  
 C \* BY AN AGENCY OF THE UNITED STATES GOVERNMENT. NEITHER \*  
 C \* THE UNITED STATES GOVERNMENT NOR ANY AGENCY THEROF, \*  
 C \* NOR ANY OF THEIR EMPLOYEES, \*  
 C \* NOR ANY OF THEIR CONTRACTORS, SUBCONTRACTORS, OR THEIR \*  
 C \* EMPLOYEES, MAKES ANY WARRANTY, EXPRESS OR IMPLIED, OR \*  
 C \* ASSUMES ANY LEGAL LIABILITY OR RESPONSIBILITY FOR THE \*  
 C \* \* \* \* \* ACCURACY, \* \* \* \* \* \*  
 C \* \* \* \* \* COMPLETENESS \* \* \* \* \* \*  
 C \* \* \* \* \* OR USEFULNESS \* \* \* \* \* \*  
 C \* \* \* \* \* OF ANY \* \* \* \* \* \*  
 C \* \* \* \* \* INFORMATION, \* \* \* \* \* \*  
 C \* \* \* \* \* APPARATUS, \* \* \* \* \* \*  
 C \* \* \* \* \* PRODUCT \* \* \* \* \* \*  
 C \* \* \* \* \* OR PROCESS \* \* \* \* \* \*  
 C \* \* \* \* \* DISCLOSED, \* \* \* \* \* \*  
 C \* \* \* \* \* OR REPRESENTS \* \* \* \* \* \*  
 C \* \* \* \* \* \*\* THAT ITS \*\* \* \* \* \* \* \*  
 C \* \* \* \* \* \*\* USE WOULD NOT \*\* \* \* \* \* \* \*  
 C \* \* \* \* \* \*\* INFRINGE \*\* \* \* \* \* \* \*  
 C \* \* \* \* \* \*\* PRIVATELY \*\* \* \* \* \* \* \*  
 C \* \* \* \* \* \*\* OWNED \*\* \* \* \* \* \* \*  
 C \* \* \* \* \* \*\* RIGHTS. \*\* \* \* \* \* \* \*  
 C \* \* \* \* \* \*\* \* \* \* \* \* \*  
 C \* \* \* \* \* \*\* \* \* \* \* \* \*

C COMPUTES X, F, Y, AND G FOR GIVEN AA,BB,CC,DD,1/L, P, AND Q  
 C AA=RADIUS OF INNER MIRROR. BB=INNER RADIUS OF OUTER MIRROR  
 C CC=OUTER RADIUS OF OUTER MIRROR. DC=VERTICAL OFFSET AT OUTER  
 C EDGE OF OUTER MIRROR (D>0 = OUTER MIRROR "ABOVE" INNER).  
 C 1/L = INVERSE OF "FOCAL"LENGTH.  
 C WK(33+7\*NEQ), IW(33)  
 C COMMON /COMWRK/ WK(100),IW(33),INFO(15)  
 C COMMON /COMPQI/ INT,IOT,ZMX  
 C CHARACTER \*10 CHI,CHO  
 C COMMON /COMCHA/ CHI,CHO

```

COMMON PI,B,C,D,RL,E,XK,RERR,IERR
DIMENSION F(2),FP(2),CU(4)
EXTERNAL PINT,QINT,FPEV
RERR=.00001
NITER=15
PI=ACOS(-1.)
10 WRITE(5,*)' ENTER A, B, C, D, 1/L AND NP'
READ(5,*)AA,BB,CC,DD,RRL,NP
IF(AA.LE.0)STOP
B=BB/AA
C=CC/AA
D=DD/AA
RL=RRL*AA
XK=SQRT(C**2+D**2)+D
IF(NP.LT.10.OR.RERR.LE.0.OR.C.LE.B.OR.B.LT.1.01.OR.
1 RL.GE.0.1)THEN
WRITE(5,*)' INPUT ERROR'
GO TO 10
ENDIF
INT=0
IOT=0
ERR=RERR
Z1=B
Z2=C
CALL QNC3(QINT,Z1,Z2,ERR,QINTV,IE)
C QNC3 IS A ROUTINE FROM THE SANDIA LABS "MATHLIB" LIBRARY.
C QINTV - INTEGRAL FROM Z1 TO Z2 OF QINT(X), WITH ERROR = ERR.
C IE IS AN ERROR SWITCH.
IF(IE.NE.1) THEN
WRITE(5,*)' ERROR QNC3 Q',IE
STOP
ENDIF
ZM=.98/(1+RL)
IF(RL.LE.0)ZM=1.
ZMX=ZM
DO 15 ITER=1,NITER
ERR=RERR
Z1=0
Z2=ZM
CALL QNC3(PINT,Z1,Z2,ERR,PINTV,IE)
IF(IE.NE.1) THEN
WRITE(5,*)' ERROR QNC3 P',IE
STOP
ENDIF
E=PINTV/QINTV
IF(RL.LE.0)GO TO 16
DO 11 I=1,15
11 INFO(I)=0
Z1=0
Z2=ZM
F(1)=0
F(2)=C
RE=RERR
AE=RERR
INFO(1)=0

```

```

      CALL DERKF(FPEV,2,Z1,F,Z2,INFO,RE,AE,IE,WK,100,IW,33,CU,
1 IDUMMY)
C DERKF IS A RUNGE-KUTTA TYPE ROUTINE FOR THE INTEGRATION OF SYSTEMS OF
C DIFFERENTIAL EQUATIONS. IT IS IN THE SANDIA LAB'S "SLATEC" LIBRARY.
C HERE FPEV EVALUATES THE DERIVATIVES FOR THE ANSWER VECTOR F. THERE ARE
C TWO EQUATIONS IN THE SYSTEM AND THE INTEGRATION IS FROM Z1 TO Z2.
C RE AND AE ARE THE RELATIVE AND ABSOLUTE ERRORS, RESPECTIVELY. INFO,
C CU, AND IDUMMY ARE COMMUNICATION ARRAYS AND WK AND IW ARE WORK ARRAYS.
C IE IS AN ERROR SWITCH.
      IF(IE.NE.2.AND.IE.NE.-5) THEN
        WRITE(5,*)'ERROR DERKF(DOINT) ',IE,I,Z
        STOP
      ENDIF
      CALL FPEV(ZM,F,FP,CU,IDUMMY)
      ZM=ZM-(CU(1)-1.)/(ZM*FP(1)*RL+F(1)*RL+1)
      ZMX=ZM
      IF(ABS(CU(1)-1.).LE.RERR)GO TO 16
15 CONTINUE
      WRITE(5,*)' ERROR - NOT CONVERGE IN ',NITER
      STOP
16 Z1=AA*ZM
      WRITE(6,1)AA,BB,CC,DD,RRL,NP,RERR,Z1,ITER,CHI,CHO
1  FORMAT('LWXIO      A      B      C      D      1/L      NP'/
1 5X,4F8.4,F9.4,I5/9X,E12.3,F9.4,I4/1X,A10,'-',A10/
2 11X,'X',11X,'F',11X,'Y',11X,'G',7X,'DF/DX',7X,'DG/DY')
      CALL FINAL(NP,AA,ZM)
      GO TO 10
      END
C -----
      SUBROUTINE FINAL(NP,AA,ZM)
      COMMON /COMWRK/ WK(100),IW(33),INFO(15)
      COMMON PI,B,C,D,RL,E,XK,RERR,IERR
      DIMENSION F(2),FP(2),CU(4)
C  CU = X, G, DF/DX, DG/DY
      EXTERNAL FPEV
      DO 10 I=1,15
10  INFO(I)=0
      Z=0
      F(1)=0
      F(2)=C
      CALL FPEV(Z,F,FP,CU,IDUMMY)
      PRX=AA*CU(1)
      PRF=AA*F(1)
      PRY=AA*F(2)
      PRG=AA*CU(2)
      IF(D.LT.0) THEN
        PRF=PRF-D*AA
        PRG=PRG-D*AA
      ENDIF
      WRITE(6,1)PRX,PRF,PRY,PRG,CU(3),CU(4)
      DZ=ZM/FLOAT(NP-1)
      DO 20 I=2,NP
      ZEND=(I-1)*DZ
      IF(I.EQ.NP)ZEND=ZM
      RE=RERR

```

```

AE=RERR
INFO(1)=0
CALL DERKF(FPEV,2,Z,F,ZEND,INFO,RE,AE,IE,WK,100,IW,33,CU,
1 IDUMMY)
IF(IE.NE.2.AND.IE.NE.-5) THEN
  WRITE(5,*)'ERROR DERKF ',IE,I,Z
  STOP
ENDIF
Z=ZEND
CALL FPEV(Z,F,FP,CU, IDUMMY)
PRX=AA*CU(1)
PRF=AA*F(1)
PRY=AA*F(2)
PRG=AA*CU(2)
IF(D.LT.0) THEN
  PRF=PRF-D*AA
  PRG=PRG-D*AA
ENDIF
WRITE(6,1)PRX,PRF,PRY,PRG, CU(3), CU(4)
1 FORMAT(1X,6F12.5)
20 CONTINUE
RETURN
END

C -----
SUBROUTINE FPEV(Z,F,FP,CU, IDUMMY)
DIMENSION F(2),FP(2),CU(4)
C CU = X, G, DF/DX, DG/DY
COMMON PI,B,C,D,RL,E,XK,RERR,IERR
FLP=RL*F(1)+1
X=Z*FLP
TTH=Z*RL
Z1=TTH**2
H=(XK-2*F(1)-X*TTH*(((.02734375*Z1-.0390625)*Z1+.0625)*
1 Z1-.125)*Z1+.5))/(F(2)-X)
S=.5*(H**2-1)/H
Z2=ATAN2(S*TTH+1,TTH-S)
IF(Z2.LT.0)Z2=Z2+PI
Z2=TAN(Z2/2)
FP(1)=Z2*FLP/(1-Z2*TTH)
FP(2)=-Z*PEV(Z)/(E*F(2)*QEV(F(2)))
CU(1)=X
CU(2)=S*(F(2)-X)+F(1)
DXDZ=TTH*FP(1)+FLP
CU(3)=FP(1)/DXDZ
IF(Z.LE.0)THEN
  CU(4)=-1/CU(3)
ELSE
  Z3=(RL*X*DXDZ+FLP*FP(1))/SQRT((RL*X)**2+FLP**2)
  Z4=(H**2+1)*(H*DXDZ-H*FP(2)-FP(1)-Z3)/(2*H*H*(F(2)-X))
  CU(4)=(Z4*(F(2)-X)+S*(FP(2)-DXDZ)+FP(1))/FP(2)
ENDIF
RETURN
END

C -----
FUNCTION PINT(X)

```

```
PINT=X*PEV(X)
RETURN
END
```

```
C -----
FUNCTION QINT(X)
QINT=X*QEV(X)
RETURN
END
```

```
C -----
FUNCTION PEV(X)
PARAMETER (MXINT=5)
CHARACTER *10 CHI,CHO,CH(MXINT)
COMMON /COMPQI/ INT,IOT,ZMX
COMMON /COMCHA/ CHI,CHO
DATA CH/'UNIFORM','COSINE','GAUSS-1','GAUSS-2','GAUSS-3'/
IF(INT.EQ.0) THEN
90  WRITE(5,91)(I,CH(I),I=1,MXINT)
91  FORMAT(' ENTER INPUT DENSITY'/(5(1X,I3,'-',A10)))
  READ(5,*)INT
  IF(INT.LT.1.OR.INT.GT.MXINT)GO TO 90
  CHI=CH(INT)
  PI2=ACOS(-1.)/2
ENDIF
GO TO (1,2,3,4,5),INT
1  PEV=1.
  RETURN
2  PEV=COS(PI2*X/ZMX)
  RETURN
3  PEV=EXP(-.5*((X/ZMX)**2))
  RETURN
4  PEV=EXP(-2.*((X/ZMX)**2))
  RETURN
5  PEV=EXP(-4.5*((X/ZMX)**2))
  RETURN
END
```

```
C -----
FUNCTION QEV(X)
PARAMETER (MXIOT=1)
COMMON /COMPQI/ INT,IOT,ZMX
CHARACTER *10 CHI,CHO,CH(MXIOT)
COMMON /COMCHA/ CHI,CHO
DATA CH/'UNIFORM'/
IF(IOT.EQ.0) THEN
10  WRITE(5,1)(I,CH(I),I=1,MXIOT)
1  FORMAT(' ENTER OUTPUT DENSITY SHAPE'/(1X,I3,'-',A10))
  READ(5,*)IOT
  IF(IOT.LT.1.OR.IOT.GT.MXIOT)GO TO 10
  CHO=CH(IOT)
ENDIF
IF(IOT.EQ.1)QEV=1
RETURN
END
```

DISTRIBUTION:

National Energy Software Center (12)  
Argonne National Laboratory  
9700 South Cass Avenue  
Argonne, Illinois 60439

2600 R. J. Detry  
2610 D. C. Jones  
2630 E. K. Montoya  
2640 E. J. Theriot  
2641 M. R. Scott  
2642 P. A. Lemke  
2644 R. E. Jones  
2645 W. F. Mason  
2646 L. A. Bertram  
2646 E. A. Aronson (5)  
2646 L. A. Romero (2)  
2646 A. H. Treadway (2)  
5320 G. H. Mauth  
5321 J. L. R. Williams  
5322 L. S. Walker  
5323 D. A. Reynolds  
5323 F. E. Thompson  
5324 L. J. Ellis  
8024 M. A. Pound (2)  
3141-1 C. M. Ostrander (5)  
3151 W. L. Gardner (3)

For DOE/TIC

DOE/TIC (28)

(C. Dalin, 3154-3)